```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the commutative polynomial ring $A = \mathbb{Q}(\eta_1, \eta_2)[\mathrm{d}, \sigma_1, \sigma_2]$ of differential time-delay operators, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d,t], dual_shift=[sigma[1],s1],
> dual_shift=[sigma[2],s2], polynom=[t,s1,s2], comm=[eta[1],eta[2]],
> shift_action=[sigma[1],t,h[1]], shift_action=[sigma[2],t,h[2]]):
```

where $\mathrm{d}\ \mathrm{y}(t) = \frac{d}{dt}\,\mathrm{y}(t)$ and $\sigma_i\ \mathrm{y}(t) = \mathrm{y}(t - h_i)$ for i = 1, 2, the matrix P with entries in A defined by

```
> P := evalm([[1,1,-1,-1,0,0],[d+eta[1],d-eta[1],-eta[2],eta[2],0,0],
> [sigma[1]^2,1,0,0,-sigma[1],0],[0,0,1,sigma[2]^2,0,-sigma[2]]]);
```

$$P := \begin{bmatrix} 1 & 1 & -1 & -1 & 0 & 0 \\ d+\eta_1 & d-\eta_1 & -\eta_2 & \eta_2 & 0 & 0 \\ {\sigma_1}^2 & 1 & 0 & 0 & -\sigma_1 & 0 \\ 0 & 0 & 1 & {\sigma_2}^2 & 0 & -\sigma_2 \end{bmatrix}$$

and the A-module L finitely presented by P which defines a vibrating string with an interior mass considered in H. Mounier, J. Rudolph, M. Fliess, P. Rouchon, "Tracking control of a vibrating string with an interior mass viewed as a delay system", ESAIM Control, Optimisation and Calculus of Variations, 3 (1998), pp. 315-321. Let us apply Stafford's reduction to L.

```
> S := StaffordReduction(P,A,"reduce_relations"=true);
```

$$S := \left[\begin{bmatrix} -2\,\eta_1 & d+\eta_1-\eta_2 & d+\eta_1+\eta_2 & 0 & 0 \\ -1+{\sigma_1}^2 & -{\sigma_1}^2 & -{\sigma_1}^2 & \sigma_1 & 0 \\ 0 & 1 & {\sigma_2}^2 & 0 & -\sigma_2 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}\right]$$

We find that the A-module L_2 finitely presented by S[1] is isomorphic to L, and the A-homomorphism γ from L_2 to L induced by S[2] is an A-isomorphism:

```
> TestIso(S[1],P,S[2],A);
```

true

Let us try to apply Stafford's reduction to the presentation of L_2.

```
> T := StaffordReduction(S[1],A,"reduce_relations"=true):
```

We obtain that L_2 is isomorphic to the A-module L_3 finitely presented by T[1], where T[1] is defined by

```
> T[1];
```

$$\left[\begin{array}{c} (-\eta_1-\eta_2)\,{\sigma_1}^2+{\sigma_1}^2\,d-d-\eta_1+\eta_2\,,\ (-\eta_1+\eta_2)\,{\sigma_1}^2+{\sigma_1}^2\,d-\eta_1-\eta_2-d\,,\ 2\,\eta_1\,\sigma_1\,,\ 0 \\ 1\,,\ {\sigma_2}^2\,,\ 0\,,\ -\sigma_2 \end{array}\right]$$

and the A-homomorphism β from L_3 to L_2 induced by T[2], where T[2] is defined by

```
> T[2];
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

is an A-isomorphism:

```
> TestIso(T[1],S[1],T[2],A);
```

$$\mathit{true}$$

Let us try again to apply Stafford's reduction to L_3:

```
> U := map(collect,StaffordReduction(T[1],A,"reduce_relations"=true),
> [sigma[1],sigma[2],d],distributed);
```

$$U := \Big[\Big[d\,\sigma_1{}^2\,\sigma_2{}^2 - \sigma_1{}^2\,d - d\,\sigma_2{}^2 + (-\eta_1 - \eta_2)\,\sigma_1{}^2\,\sigma_2{}^2 + (-\eta_1 + \eta_2)\,\sigma_2{}^2 + (\eta_1 - \eta_2)\,\sigma_1{}^2 + \eta_1 + \eta_2 + d,\ -2\,\eta_1\,\sigma_1\,,\ d\,\sigma_2 + (\eta_1 - \eta_2)\,\sigma_2 + (\eta_1 + \eta_2)\,\sigma_1{}^2\,\sigma_2 - \sigma_1{}^2\,d\,\sigma_2\Big],\ \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}\Big]$$

```
> rowdim(U[1]); coldim(U[1]);
```

$$1$$
$$3$$

We obtain that L_3 is isomorphic to the A-module L' finitely presented by U[1] = (U[1][1,1] U[1][1,2] U[1][1,3]) and

```
> U[1][1,1];
```

$$d\,\sigma_1{}^2\,\sigma_2{}^2 - \sigma_1{}^2\,d - d\,\sigma_2{}^2 + (-\eta_1 - \eta_2)\,\sigma_1{}^2\,\sigma_2{}^2 + (-\eta_1 + \eta_2)\,\sigma_2{}^2 + (\eta_1 - \eta_2)\,\sigma_1{}^2 + \eta_1 + \eta_2 + d$$

```
> U[1][1,2];
```

$$-2\,\eta_1\,\sigma_1$$

```
> U[1][1,3];
```

$$d\,\sigma_2 + (\eta_1 - \eta_2)\,\sigma_2 + (\eta_1 + \eta_2)\,\sigma_1{}^2\,\sigma_2 - \sigma_1{}^2\,d\,\sigma_2$$

and the A-homomorphism α from L' to L_3 induced by U[2], where U[2] is defined by

```
> U[2];
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

is an A-isomorphism:

```
> TestIso(U[1],T[1],U[2],A);
```

$$\mathit{true}$$

If we define W = U[2] T[2] S[2], namely,

```
> W := Mult(U[2],T[2],S[2],A);
```

$$\mathrm{W} := \begin{bmatrix} 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

then the A-homomorphism $\theta = \gamma$ o β o α from L' to L induced by W is an A-isomorphism:

```
> TestIso(U[1],P,W,A);
```

true

Performing algebraic simplification on V[1] = U[1]ˆT, namely

```
> with(PurityFiltration):
> V := ReducedPresentation(transpose(U[1]),A):
> V[1];
```

$$\Big[d\,\sigma_1{}^2\,\sigma_2{}^2 - \sigma_1{}^2\,d - d\,\sigma_2{}^2 + (-\eta_1 - \eta_2)\,\sigma_1{}^2\,\sigma_2{}^2 + (-\eta_1 + \eta_2)\,\sigma_2{}^2 + (\eta_1 - \eta_2)\,\sigma_1{}^2 + \eta_1 + \eta_2 + d\Big]$$

$$[-2\,\eta_1\,\sigma_1]$$

$$[d\,\sigma_2 + (\eta_1 - \eta_2)\,\sigma_2 + (\eta_1 + \eta_2)\,\sigma_1{}^2\,\sigma_2 - \sigma_1{}^2\,d\,\sigma_2]$$

we get that the A-modules finitely presented by V[1] and V[2], respectively, are isomorphic, where V[2] is defined by

```
> V[2];
```

$$\begin{bmatrix} 2\,\eta_1\,\sigma_1 \\ 2\,\eta_1\,\eta_2\,\sigma_2 \\ 2\,\eta_1\,d + 2\,\eta_1{}^2 + 2\,\eta_1\,\eta_2 \end{bmatrix}$$

the corresponding A-isomorphism being defined by the identity map, i.e.:

```
> V[3];
```

$$\begin{bmatrix} 1 \end{bmatrix}$$

This result can be checked again:

```
> TestIso(V[1],V[2],V[3],A);
```

true

If we denote by Q a matrix satisfying V[2] = QˆT V[1], namely

```
> Q := transpose(Factorize(V[2],V[1],A));
```

$$Q := \begin{bmatrix} 0 & \eta_1\,\sigma_2 & 2\,\eta_1 \\ -1 & -\sigma_2\,\sigma_1\,\eta_2 & (\eta_1 - \eta_2)\,\sigma_1 - d\,\sigma_1 \\ 0 & -\eta_1 + \eta_1\,\sigma_2{}^2 & 2\,\eta_1\,\sigma_2 \end{bmatrix}$$

then we have V[2]ˆT = V[1]ˆT Q = U[1] Q, and we get that L' is isomorphic to the A-module M finitely presented by V[2], or, equivalently, finitely presented by R = V[2]ˆT / (2 η_1), namely,

```
> R := simplify(evalm(transpose(V[2])/(2*eta[1])));
```

$$R := \left[\begin{array}{ccc} \sigma_1 & \eta_2\,\sigma_2 & d+\eta_1+\eta_2 \end{array}\right]$$

and the A-isomorphism ω from L' onto M is induced by Q. Let us check again that ω is an A-isomorphism:

```
> TestIso(U[1],R,Q,A);
```

$$\mathit{true}$$

Then, ω^{-1} is induced by X[1], where X[1] is defined by

```
> X := InverseMorphism(U[1],R,Q,A);
```

$$X := \left[\left[\begin{array}{ccc} \frac{1}{2}\,\frac{\sigma_1\,(\eta_1-\eta_2-d-\eta_1\,{\sigma_2}^2-\eta_2\,{\sigma_2}^2+d\,{\sigma_2}^2)}{\eta_1} & -1 & -\frac{1}{2}\,\frac{\sigma_1\,\sigma_2\,(-\eta_1-\eta_2+d)}{\eta_1} \\ \frac{\sigma_2}{\eta_1} & 0 & -\frac{1}{\eta_1} \\ -\frac{1}{2}\,\frac{{\sigma_2}^2-1}{\eta_1} & 0 & \frac{1}{2}\,\frac{\sigma_2}{\eta_1} \end{array}\right], \left[\begin{array}{c} \frac{1}{2}\,\frac{1}{\eta_1} \end{array}\right]\right]$$

Now, if we define Z = X[1] W, namely,

```
> Z := Mult(X[1],W,A);
```

$$Z := \left[\begin{array}{cccccc} 0 & 0 & 0 & \frac{1}{2}\,\frac{\sigma_1\,(\eta_1-\eta_2-d-\eta_1\,{\sigma_2}^2-\eta_2\,{\sigma_2}^2+d\,{\sigma_2}^2)}{\eta_1} & -1 & -\frac{1}{2}\,\frac{\sigma_1\,\sigma_2\,(-\eta_1-\eta_2+d)}{\eta_1} \\ 0 & 0 & 0 & \frac{\sigma_2}{\eta_1} & 0 & -\frac{1}{\eta_1} \\ 0 & 0 & 0 & -\frac{1}{2}\,\frac{{\sigma_2}^2-1}{\eta_1} & 0 & \frac{1}{2}\,\frac{\sigma_2}{\eta_1} \end{array}\right]$$

then the A-homomorphism $\rho = \omega^{-1}$ o θ from M to L is the A-isomorphism induced by Z

```
> TestIso(R,P,Z,A);
```

$$\mathit{true}$$

which shows that the linear differential time-delay system defining the string model with an interior mass is equivalent to the single differential time-delay equation:

$$\tfrac{d}{dt}\,z_3(t) + (\eta_1+\eta_2) z_3(t) + z_1(t-h_1) + \eta_2\, z_2(t-h_2) = 0.$$

Finally, ρ^{-1} is induced by Y[1], where Y[1] is defined by

```
> Y := InverseMorphism(R,P,Z,A);
```

$$Y := \left[\left[\begin{array}{ccc} 0 & -\eta_2\,\sigma_2 & -\eta_2+\eta_1-d \\ 0 & \eta_2\,\sigma_2 & d+\eta_1+\eta_2 \\ 0 & -\eta_1\,\sigma_2 & 0 \\ 0 & \eta_1\,\sigma_2 & 2\,\eta_1 \\ -1 & -\sigma_2\,\sigma_1\,\eta_2 & -d\,\sigma_1+\eta_1\,\sigma_1-\sigma_1\,\eta_2 \\ 0 & -\eta_1+\eta_1\,{\sigma_2}^2 & 2\,\eta_1\,\sigma_2 \end{array}\right], \left[\begin{array}{c} 0 \\ 0 \\ 1 \\ 0 \end{array}\right]\right]$$

a fact which can be checked again:

```
> TestIso(P,R,Y[1],A);
```

$$\mathit{true}$$